```
> with(plots):with(linalg):
> F:=[3*y,4*z,-6*x]:vars:=[x,y,z]:cF:=curl(F,
  vars);
```

$$cF := [-4, 6, -3]$$

```
> zs:=9-x^2-y^2:plot3d(zs,x=-3..3,y=-3..3,vie
  w=0..9,axes=BOXED);
```



Bounds on x, y, and z are used in defining this graph. The intersection with z=0 is the circle x^2+y^2=9, which should be parametrized with trigonometric functions. The line integral is easiest.

```
> xt:=3*cos(t);yt:=3*sin(t);zt:=0;Fb:=subs(x=
  xt,y=yt,z=zt,F);
```

$$xt := 3\cos(t)$$

$$yt := 3\sin(t)$$

$$zt := 0$$

$$Fb := [9\sin(t), 0, -18\cos(t)]$$

An abuse of notation follows: the names dx, dy, dz will be used for the derivatives with respect to t.

```
> dx:=diff(xt,t);dy:=diff(yt,t);dz:=diff(zt,t
  );dr:=[dx,dy,dz];
```

$$dx := -3\sin(t)$$

$$dy := 3\cos(t)$$

$$dz := 0$$

$$dr := [-3\sin(t), 3\cos(t), 0]$$

```
> int(dotprod(Fb,dr,'orthogonal'),t=0..2*Pi);
```

$$-27\pi$$

We now use the formula for the normal to the graph of a function.

```
> zx:=diff(zs,x):zy:=diff(zs,y):Ns:=[-zx,-zy,
  1];
```

$$Ns := [2x, 2y, 1]$$

We integrate over the surface by expressing the integral over the projection into the xy plane.

```
> int(int(dotprod(cF,Ns,'orthogonal'),y=-sqrt
  (9-x^2)..sqrt(9-x^2)),x=-3..3);
```

$$-27\pi$$

Looking at the steps:

```
> int(dotprod(cF,Ns,'orthogonal'),y=-sqrt(9-x
  ^2)..sqrt(9-x^2));
```

$$-6\sqrt{9-x^2}-16\,x\sqrt{9-x^2}$$

```
> int(%,x=-3..3);
```

$$-27\,\pi$$

The polar form is (including the change-of-variables factor of r):

```
> polf:=r*subs(x=r*cos(theta),y=r*sin(theta),
  dotprod(cF,Ns,'orthogonal'));
```

$$polf := r\,(-3-8\,r\cos(\theta)+12\,r\sin(\theta))$$

```
> int(polf,r=0..3);
```

$$-72\cos(\theta)+108\sin(\theta)-\frac{27}{2}$$

```
> int(%,theta=0..2*Pi);
```

$$-27\,\pi$$

Although we usually integrate with respect to r first, this time it is usuful to do the integral in the other order.

```
> int(polf,theta=0..2*Pi);
```

$$-6\,r\,\pi$$

```
> int(%,r=0..3);
```

$$-27\,\pi$$

Of course, all calculations give the same answer.